// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (security/ReentrancyGuard.sol)

// pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        // On the first call to nonReentrant, _notEntered will be true
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;

        _;

        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }
}


// OpenZeppelin Contracts v4.4.1 (access/IAccessControl.sol)

// pragma solidity ^0.8.0;

/**
 * @dev External interface of AccessControl declared to support ERC165 detection.
 */
interface IAccessControl {
    /**
     * @dev Emitted when `newAdminRole` is set as ``role``'s admin role, replacing `previousAdminRole`
     *
     * `DEFAULT_ADMIN_ROLE` is the starting admin for all roles, despite
     * {RoleAdminChanged} not being emitted signaling this.
     *
     * _Available since v3.1._
     */
    event RoleAdminChanged(bytes32 indexed role, bytes32 indexed previousAdminRole, bytes32 indexed newAdminRole);

    /**
     * @dev Emitted when `account` is granted `role`.
     *
     * `sender` is the account that originated the contract call, an admin role
     * bearer except when using {AccessControl-_setupRole}.
     */
    event RoleGranted(bytes32 indexed role, address indexed account, address indexed sender);

    /**
     * @dev Emitted when `account` is revoked `role`.
     *
     * `sender` is the account that originated the contract call:
     *   - if using `revokeRole`, it is the admin role bearer
     *   - if using `renounceRole`, it is the role bearer (i.e. `account`)
     */
    event RoleRevoked(bytes32 indexed role, address indexed account, address indexed sender);

    /**
     * @dev Returns `true` if `account` has been granted `role`.
     */
    function hasRole(bytes32 role, address account) external view returns (bool);

    /**
     * @dev Returns the admin role that controls `role`. See {grantRole} and
     * {revokeRole}.
     *
     * To change a role's admin, use {AccessControl-_setRoleAdmin}.
     */
    function getRoleAdmin(bytes32 role) external view returns (bytes32);

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function grantRole(bytes32 role, address account) external;

    /**
     * @dev Revokes `role` from `account`.
     *
     * If `account` had been granted `role`, emits a {RoleRevoked} event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function revokeRole(bytes32 role, address account) external;

    /**
     * @dev Revokes `role` from the calling account.
     *
     * Roles are often managed via {grantRole} and {revokeRole}: this function's
     * purpose is to provide a mechanism for accounts to lose their privileges
     * if they are compromised (such as when a trusted device is misplaced).
     *
     * If the calling account had been granted `role`, emits a {RoleRevoked}
     * event.
     *
     * Requirements:
     *
     * - the caller must be `account`.
     */
    function renounceRole(bytes32 role, address account) external;
}


// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

// pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}


// OpenZeppelin Contracts v4.4.1 (utils/Strings.sol)

// pragma solidity ^0.8.0;

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _HEX_SYMBOLS = "0123456789abcdef";

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        // Inspired by OraclizeAPI's implementation - MIT licence
        // https://github.com/oraclize/ethereum-api/blob/b42146b063c7d6ee1358846c198246239e9360e8/oraclizeAPI_0.4.25.sol

        if (value == 0) {
            return "0";
        }
        uint256 temp = value;
        uint256 digits;
        while (temp != 0) {
            digits++;
            temp /= 10;
        }
        bytes memory buffer = new bytes(digits);
        while (value != 0) {
            digits -= 1;
            buffer[digits] = bytes1(uint8(48 + uint256(value % 10)));
            value /= 10;
        }
        return string(buffer);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        if (value == 0) {
            return "0x00";
        }
        uint256 temp = value;
        uint256 length = 0;
        while (temp != 0) {
            length++;
            temp >>= 8;
        }
        return toHexString(value, length);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _HEX_SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }
}


// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC165.sol)

// pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}


// OpenZeppelin Contracts v4.4.1 (utils/introspection/ERC165.sol)

// pragma solidity ^0.8.0;

/**
 * @dev Implementation of the {IERC165} interface.
 *
 * Contracts that want to implement ERC165 should inherit from this contract and override {supportsInterface} to check
 * for the additional interface id that will be supported. For example:
 *
 * ```solidity
 * function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
 *     return interfaceId == type(MyInterface).interfaceId || super.supportsInterface(interfaceId);
 * }
 * ```
 *
 * Alternatively, {ERC165Storage} provides an easier to use but more expensive implementation.
 */
abstract contract ERC165 is IERC165 {
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IERC165).interfaceId;
    }
}


// OpenZeppelin Contracts (last updated v4.6.0) (access/AccessControl.sol)

// pragma solidity ^0.8.0;

/**
 * @dev Contract module that allows children to implement role-based access
 * control mechanisms. This is a lightweight version that doesn't allow enumerating role
 * members except through off-chain means by accessing the contract event logs. Some
 * applications may benefit from on-chain enumerability, for those cases see
 * {AccessControlEnumerable}.
 *
 * Roles are referred to by their `bytes32` identifier. These should be exposed
 * in the external API and be unique. The best way to achieve this is by
 * using `public constant` hash digests:
 *
 * ```
 * bytes32 public constant MY_ROLE = keccak256("MY_ROLE");
 * ```
 *
 * Roles can be used to represent a set of permissions. To restrict access to a
 * function call, use {hasRole}:
 *
 * ```
 * function foo() public {
 *     require(hasRole(MY_ROLE, msg.sender));
 *     ...
 * }
 * ```
 *
 * Roles can be granted and revoked dynamically via the {grantRole} and
 * {revokeRole} functions. Each role has an associated admin role, and only
 * accounts that have a role's admin role can call {grantRole} and {revokeRole}.
 *
 * By default, the admin role for all roles is `DEFAULT_ADMIN_ROLE`, which means
 * that only accounts with this role will be able to grant or revoke other
 * roles. More complex role relationships can be created by using
 * {_setRoleAdmin}.
 *
 * WARNING: The `DEFAULT_ADMIN_ROLE` is also its own admin: it has permission to
 * grant and revoke this role. Extra precautions should be taken to secure
 * accounts that have been granted it.
 */
abstract contract AccessControl is Context, IAccessControl, ERC165 {
    struct RoleData {
        mapping(address => bool) members;
        bytes32 adminRole;
    }

    mapping(bytes32 => RoleData) private _roles;

    bytes32 public constant DEFAULT_ADMIN_ROLE = 0x00;

    /**
     * @dev Modifier that checks that an account has a specific role. Reverts
     * with a standardized message including the required role.
     *
     * The format of the revert reason is given by the following regular expression:
     *
     *  /^AccessControl: account (0x[0-9a-f]{40}) is missing role (0x[0-9a-f]{64})$/
     *
     * _Available since v4.1._
     */
    modifier onlyRole(bytes32 role) {
        _checkRole(role);
        _;
    }

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IAccessControl).interfaceId || super.supportsInterface(interfaceId);
    }

    /**
     * @dev Returns `true` if `account` has been granted `role`.
     */
    function hasRole(bytes32 role, address account) public view virtual override returns (bool) {
        return _roles[role].members[account];
    }

    /**
     * @dev Revert with a standard message if `_msgSender()` is missing `role`.
     * Overriding this function changes the behavior of the {onlyRole} modifier.
     *
     * Format of the revert message is described in {_checkRole}.
     *
     * _Available since v4.6._
     */
    function _checkRole(bytes32 role) internal view virtual {
        _checkRole(role, _msgSender());
    }

    /**
     * @dev Revert with a standard message if `account` is missing `role`.
     *
     * The format of the revert reason is given by the following regular expression:
     *
     *  /^AccessControl: account (0x[0-9a-f]{40}) is missing role (0x[0-9a-f]{64})$/
     */
    function _checkRole(bytes32 role, address account) internal view virtual {
        if (!hasRole(role, account)) {
            revert(
                string(
                    abi.encodePacked(
                        "AccessControl: account ",
                        Strings.toHexString(uint160(account), 20),
                        " is missing role ",
                        Strings.toHexString(uint256(role), 32)
                    )
                )
            );
        }
    }

    /**
     * @dev Returns the admin role that controls `role`. See {grantRole} and
     * {revokeRole}.
     *
     * To change a role's admin, use {_setRoleAdmin}.
     */
    function getRoleAdmin(bytes32 role) public view virtual override returns (bytes32) {
        return _roles[role].adminRole;
    }

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function grantRole(bytes32 role, address account) public virtual override onlyRole(getRoleAdmin(role)) {
        _grantRole(role, account);
    }

    /**
     * @dev Revokes `role` from `account`.
     *
     * If `account` had been granted `role`, emits a {RoleRevoked} event.
     *
     * Requirements:
     *
     * - the caller must have ``role``'s admin role.
     */
    function revokeRole(bytes32 role, address account) public virtual override onlyRole(getRoleAdmin(role)) {
        _revokeRole(role, account);
    }

    /**
     * @dev Revokes `role` from the calling account.
     *
     * Roles are often managed via {grantRole} and {revokeRole}: this function's
     * purpose is to provide a mechanism for accounts to lose their privileges
     * if they are compromised (such as when a trusted device is misplaced).
     *
     * If the calling account had been revoked `role`, emits a {RoleRevoked}
     * event.
     *
     * Requirements:
     *
     * - the caller must be `account`.
     */
    function renounceRole(bytes32 role, address account) public virtual override {
        require(account == _msgSender(), "AccessControl: can only renounce roles for self");

        _revokeRole(role, account);
    }

    /**
     * @dev Grants `role` to `account`.
     *
     * If `account` had not been already granted `role`, emits a {RoleGranted}
     * event. Note that unlike {grantRole}, this function doesn't perform any
     * checks on the calling account.
     *
     * [WARNING]
     * ====
     * This function should only be called from the constructor when setting
     * up the initial roles for the system.
     *
     * Using this function in any other way is effectively circumventing the admin
     * system imposed by {AccessControl}.
     * ====
     *
     * NOTE: This function is deprecated in favor of {_grantRole}.
     */
    function _setupRole(bytes32 role, address account) internal virtual {
        _grantRole(role, account);
    }

    /**
     * @dev Sets `adminRole` as ``role``'s admin role.
     *
     * Emits a {RoleAdminChanged} event.
     */
    function _setRoleAdmin(bytes32 role, bytes32 adminRole) internal virtual {
        bytes32 previousAdminRole = getRoleAdmin(role);
        _roles[role].adminRole = adminRole;
        emit RoleAdminChanged(role, previousAdminRole, adminRole);
    }

    /**
     * @dev Grants `role` to `account`.
     *
     * Internal function without access restriction.
     */
    function _grantRole(bytes32 role, address account) internal virtual {
        if (!hasRole(role, account)) {
            _roles[role].members[account] = true;
            emit RoleGranted(role, account, _msgSender());
        }
    }

    /**
     * @dev Revokes `role` from `account`.
     *
     * Internal function without access restriction.
     */
    function _revokeRole(bytes32 role, address account) internal virtual {
        if (hasRole(role, account)) {
            _roles[role].members[account] = false;
            emit RoleRevoked(role, account, _msgSender());
        }
    }
}


// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

// pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}


// OpenZeppelin Contracts v4.4.1 (interfaces/IERC20.sol)

// pragma solidity ^0.8.0;


// OpenZeppelin Contracts (last updated v4.5.0) (utils/Address.sol)

// pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}


// OpenZeppelin Contracts v4.4.1 (token/ERC20/utils/SafeERC20.sol)

// pragma solidity ^0.8.0;

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using Address for address;

    function safeTransfer(
        IERC20 token,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    function safeTransferFrom(
        IERC20 token,
        address from,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        require(
            (value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    function safeIncreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        uint256 newAllowance = token.allowance(address(this), spender) + value;
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    function safeDecreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        unchecked {
            uint256 oldAllowance = token.allowance(address(this), spender);
            require(oldAllowance >= value, "SafeERC20: decreased allowance below zero");
            uint256 newAllowance = oldAllowance - value;
            _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
        }
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address.functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        if (returndata.length > 0) {
            // Return data is optional
            require(abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
        }
    }
}


// OpenZeppelin Contracts (last updated v4.5.0) (utils/math/Math.sol)

// pragma solidity ^0.8.0;

/**
 * @dev Standard math utilities missing in the Solidity language.
 */
library Math {
    /**
     * @dev Returns the largest of two numbers.
     */
    function max(uint256 a, uint256 b) internal pure returns (uint256) {
        return a >= b ? a : b;
    }

    /**
     * @dev Returns the smallest of two numbers.
     */
    function min(uint256 a, uint256 b) internal pure returns (uint256) {
        return a < b ? a : b;
    }

    /**
     * @dev Returns the average of two numbers. The result is rounded towards
     * zero.
     */
    function average(uint256 a, uint256 b) internal pure returns (uint256) {
        // (a + b) / 2 can overflow.
        return (a & b) + (a ^ b) / 2;
    }

    /**
     * @dev Returns the ceiling of the division of two numbers.
     *
     * This differs from standard division with `/` in that it rounds up instead
     * of rounding down.
     */
    function ceilDiv(uint256 a, uint256 b) internal pure returns (uint256) {
        // (a + b - 1) / b can overflow on addition, so we distribute.
        return a / b + (a % b == 0 ? 0 : 1);
    }
}


// Root file: contracts\KazakFarmPortal.sol

pragma solidity 0.8.4;

/**
 * #1: Users can stake one token: USDT.
 * #2: Users can receive USDT using a traditional staking system. The initial APR is 38%.
 * #3: Users can have affiliates. They receive 10% of the rewards of their affiliates (without reducing their rewards).
 * #4: Users that joined the platform using an affiliate link get an APY of 40%.
 * #5: Users can compound their rewards without initiating any actual USDT transfer, lowering gas fees.
 * #6: Reward distribution is every block.
 * #7: Funds are sent to a separate wallet. This contract can use those funds at will.
 * #8: There is a limit on the total number of tokens that can be staked at a given time. This limit can be changed by the DAO.
 */

struct UserInfo {
  uint256 balance; // Amount of staked USDT
  uint256 lastClaimed; // The day on which the user last withdrew rewards
  address affiliateParent; // Affiliate parent of this user, zero address if none
  uint256 numberOfAffiliates; // The number of affiliates this user has
  uint256 toClaimFromAffiliates; // The amount of USDT this user can claim from his affiliates
  uint256 earnedFromAffiliates; // The amount of USDT this user has earned from his affiliates
}

contract KazakFarmPortal is AccessControl, ReentrancyGuard {
  using SafeERC20 for IERC20;

  bytes32 public constant DAO = keccak256("DAO"); // DAO role
  // uint256 public constant YEAR_IN_SECONDS = 31557600; // Number of seconds in a year

  address public dao; // DAO - can call DAO methods
  address private bank; // Where the USDT is stored
  IERC20 public usdt; // USDT token
  uint256 public standardInvertedAprPerSecond; // (100 * YEAR_IN_SECONDS / standardApr)
  uint256 public affiliateInvertedAprPerSecond; // (100 * YEAR_IN_SECONDS / affiliateApr)
  uint256 public affiliateParentQuotient; // Percentage of rewards the affiliate parent receives when their affiliates claim their rewards. Stored as a quotient

  uint256 public totalStaked = 0; // Total USDT staked since the day before or earlier. Earn rewards
  uint256 public maxTotalStaked; // Maximum USDT that can be staked on the contract at one time

  bool public emergencyExitAllowed = false; // If true, enables emergency exit
  bool public rewardsPaused = true; // If true, prevents users from claiming

  mapping(address => UserInfo) public userInfo; // Stores most pool data

  // Events
  event Staked(address indexed account, uint256 amount);
  event Unstaked(address indexed account, uint256 amount);
  event ClaimedRewards(address indexed account, uint256 amount);

  constructor(
    address _bank, // The address that receives USDT. Rewards are sent from this address
    address _usdt,
    uint256 _standardInvertedAprPerSecond,
    uint256 _affiliateInvertedAprPerSecond,
    uint256 _affiliateParentQuotient,
    uint256 _maxTotalStaked
  ) {
    // DAO setup
    dao = msg.sender;
    _setupRole(DEFAULT_ADMIN_ROLE, msg.sender);
    grantRole(DAO, msg.sender);

    // Initialize contract
    bank = _bank;
    usdt = IERC20(_usdt);
    standardInvertedAprPerSecond = _standardInvertedAprPerSecond;
    affiliateInvertedAprPerSecond = _affiliateInvertedAprPerSecond;
    affiliateParentQuotient = _affiliateParentQuotient;
    maxTotalStaked = _maxTotalStaked;
  }

  // Returns available rewards for the given account
  function getRewardsToClaim(address account) public view returns (uint256) {
    // Use standard or affiliate APR
    uint256 invertedAprPerSecond = standardInvertedAprPerSecond;
    if (userInfo[account].affiliateParent != address(0)) {
      invertedAprPerSecond = affiliateInvertedAprPerSecond;
    }

    // @dev USDT has 6 decimals, underflow risk?
    return (userInfo[account].balance * (block.timestamp - userInfo[account].lastClaimed)) / invertedAprPerSecond; // * apr / 100 / YEAR_IN_SECONDS
  }

  // Stake "amount" USDT tokens
  function stake(uint256 amount, address affiliateParent) external nonReentrant {
    // Claim
    _claim(msg.sender);

    // Call internal function
    _stake(msg.sender, amount, affiliateParent);
  }

  // Claims rewards and stakes them
  function compound() external nonReentrant {
    // Call internal function
    _compound(msg.sender);
  }

  // Claims rewards
  function claim() external nonReentrant {
    // Call internal function
    _claim(msg.sender);
  }

  // Unstake USDT
  function unstake(uint256 amount) external nonReentrant {
    // Claim
    _claim(msg.sender);

    // Call internal function
    _unstake(msg.sender, amount);
  }

  // Exit: unstake all your USDT
  function exit() external nonReentrant {
    // Claim
    _claim(msg.sender);

    // Call internal function
    _exit(msg.sender);
  }

  // Emergency exit: unstake all your USDT, do not claim, do not update the pool. Usage toggled by the DAO
  // To be used if there is something wrong with the contract, so funds will not be locked & lost
  function emergencyExit() external nonReentrant {
    // Call internal function
    _emergencyExit(msg.sender);
  }

  // Stake "amount" USDT tokens for "account"
  function _stake(
    address account,
    uint256 amount,
    address affiliateParent
  ) internal {
    require(amount > 0, "Cannot stake 0");
    require(usdt.balanceOf(account) >= amount, "USDT balance is too low");
    require(affiliateParent != account, "Cannot be your own affiliate");
    require(userInfo[affiliateParent].affiliateParent != account, "Cannot be the affiliate of your affiliate");

    // Set affiliate parent if the user does not already have one
    if (userInfo[account].affiliateParent == address(0)) {
      userInfo[account].affiliateParent = affiliateParent;
      userInfo[affiliateParent].numberOfAffiliates++;
    }

    // Increase total staked amount
    totalStaked += amount;
    require(totalStaked <= maxTotalStaked, "Cannot stake more than limit allows");

    // Send USDT
    usdt.safeTransferFrom(account, bank, amount);

    // Emit event
    emit Staked(account, amount);

    // Update user info
    userInfo[account].balance += amount;
  }

  // Compounds rewards for "account"
  // Using custom logic to avoid sending USDT - directly updates the user balance instead
  function _compound(address account) internal {
    require(!rewardsPaused, "Reward distribution is paused");

    // Compute USDT to compound
    uint256 rewardsToCompound = getRewardsToClaim(account);

    // Add rewards from affiliates to total
    uint256 totalToCompound = rewardsToCompound + userInfo[account].toClaimFromAffiliates;
    require(totalToCompound > 0, "Cannot compound 0 USDT");

    // Update user info
    userInfo[account].lastClaimed = block.timestamp;
    userInfo[account].toClaimFromAffiliates = 0;

    // Affiliate parent
    address affiliateParent = userInfo[account].affiliateParent;

    // Affiliate parent
    if (affiliateParent != address(0)) {
      // Compute affiliate parent rewards
      uint256 affiliateParentRewards = rewardsToCompound / affiliateParentQuotient;

      // Add the rewards to the parent's userInfo
      userInfo[affiliateParent].toClaimFromAffiliates += affiliateParentRewards;
      userInfo[affiliateParent].earnedFromAffiliates += affiliateParentRewards;
    }

    // Increase total staked amount
    totalStaked += totalToCompound;
    require(totalStaked <= maxTotalStaked, "Cannot stake more than limit allows");

    // Compound USDT rewards + rewards earned from affiliates
    userInfo[account].balance += totalToCompound;

    // Emit event
    emit ClaimedRewards(account, totalToCompound);
    emit Staked(account, totalToCompound);
  }

  // Claims rewards for "account"
  function _claim(address account) internal {
    require(!rewardsPaused, "Reward distribution is paused");

    // Compute USDT to claim
    uint256 rewardsToClaim = getRewardsToClaim(account);

    // Add rewards from affiliates to total
    uint256 totalToClaim = rewardsToClaim + userInfo[account].toClaimFromAffiliates;

    // Update user info
    userInfo[account].lastClaimed = block.timestamp;
    userInfo[account].toClaimFromAffiliates = 0;

    // Affiliate parent
    address affiliateParent = userInfo[account].affiliateParent;

    if (totalToClaim > 0) {
      // Affiliate parent
      if (affiliateParent != address(0)) {
        // Compute affiliate parent rewards
        uint256 affiliateParentRewards = rewardsToClaim / affiliateParentQuotient;

        // Add the rewards to the parent's userInfo
        userInfo[affiliateParent].toClaimFromAffiliates += affiliateParentRewards;
        userInfo[affiliateParent].earnedFromAffiliates += affiliateParentRewards;
      }

      // Send USDT rewards + rewards earned from affiliates
      usdt.safeTransferFrom(bank, account, totalToClaim);
    }

    // Emit event
    emit ClaimedRewards(account, totalToClaim);
  }

  // Unstake USDT
  function _unstake(address account, uint256 amount) internal {
    // Safety checks
    require(amount > 0, "Cannot unstake 0 USDT");
    require(userInfo[account].balance >= amount, "User staked USDT balance is too low");

    // Update user info
    userInfo[account].balance -= amount;
    totalStaked -= amount;

    // Emit event
    emit Unstaked(account, amount);

    // Then send the amount
    usdt.safeTransferFrom(bank, account, amount);
  }

  // Exit: unstake all your USDT
  function _exit(address account) internal {
    // Unstake all of "account"'s balance
    _unstake(account, userInfo[account].balance);
  }

  // Emergency exit: unstake all your USDT, do not claim, do not update the pool. Usage toggled by the DAO
  // To be used if there is something wrong with the contract, so funds will not be locked & lost
  function _emergencyExit(address account) internal {
    // Safety checks
    require(emergencyExitAllowed, "Emergency exit is not allowed");

    // Unstake all of "account"'s balance without doing anything else in the contract
    uint256 amount = userInfo[account].balance;

    userInfo[account].balance = 0;

    usdt.safeTransferFrom(bank, account, amount);
  }

  // Stake "amount" USDT tokens for "account"
  function stakeDAO(
    address account,
    uint256 amount,
    address affiliateParent
  ) external onlyRole(DAO) {
    // Claim
    _claim(account);

    // Call internal function
    _stake(account, amount, affiliateParent);
  }

  // Claims rewards and stakes them
  function compoundDAO(address account) external onlyRole(DAO) {
    // Compound
    _compound(account);
  }

  // Claims rewards for "account"
  function claimDAO(address account) external onlyRole(DAO) {
    // Call internal function
    _claim(account);
  }

  // Unstake USDT
  function unstakeDAO(address account, uint256 amount) external onlyRole(DAO) {
    // Claim
    _claim(account);

    // Call internal function
    _unstake(account, amount);
  }

  // Exit: unstake all your USDT
  function exitDAO(address account) external onlyRole(DAO) {
    // Claim
    _claim(account);

    // Call internal function
    _exit(account);
  }

  // Emergency exit: unstake all your USDT, do not claim, do not update the pool. Usage toggled by the DAO
  // To be used if there is something wrong with the contract, so funds will not be locked & lost
  function emergencyExitDAO(address account) external onlyRole(DAO) {
    // Call internal function
    _emergencyExit(account);
  }

  // Toggles emergency exit. Should only be used if something is wrong with the contract
  function toggleEmergencyExitDAO(bool _emergencyExitAllowed) external onlyRole(DAO) {
    emergencyExitAllowed = _emergencyExitAllowed;
  }

  // Withdraws an amount of BNB stored on the contract
  function withdrawDAO(uint256 _amount) external onlyRole(DAO) {
    payable(msg.sender).transfer(_amount);
  }

  // Withdraws an amount of ERC20 tokens stored on the contract
  function withdrawERC20DAO(address _erc20, uint256 _amount) external onlyRole(DAO) {
    IERC20(_erc20).transfer(msg.sender, _amount);
  }

  // Changes the bank address
  function changeBankDAO(address _bank) external onlyRole(DAO) {
    bank = _bank;
  }

  // Pauses / unpauses reward distribution
  function pauseRewardsDAO(bool _rewardsPaused) external onlyRole(DAO) {
    rewardsPaused = _rewardsPaused;
  }

  // Changes the limit on the number of USDT that can be staked
  function changeMaxTotalStakedDAO(uint256 _maxTotalStaked) external onlyRole(DAO) {
    maxTotalStaked = _maxTotalStaked;
  }

  // Used to change the standard APR
  function changeStandardInvertedAprPerSecondDAO(uint256 _standardInvertedAprPerSecond) external onlyRole(DAO) {
    standardInvertedAprPerSecond = _standardInvertedAprPerSecond;
  }

  // Used to change the affiliate APR
  function changeAffiliateInvertedAprPerSecondDAO(uint256 _affiliateInvertedAprPerSecond) external onlyRole(DAO) {
    affiliateInvertedAprPerSecond = _affiliateInvertedAprPerSecond;
  }

  // Used to change the percentage of rewards users earn from their affiliates
  function changeAffiliateParentQuotientDAO(uint256 _affiliateParentQuotient) external onlyRole(DAO) {
    affiliateParentQuotient = _affiliateParentQuotient;
  }

  // Change DAO
  function changeDAO(address _dao) external onlyRole(DAO) {
    revokeRole(DAO, dao);
    grantRole(DAO, _dao);
    dao = _dao;
  }

  // Revoke DAO
  function revokeDAO(address _daoToRevoke) external onlyRole(DAO) {
    revokeRole(DAO, _daoToRevoke);
  }
}